

NAG Fortran Library Routine Document

G01EFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01EFF returns the lower or upper tail probability of the gamma distribution, with parameters α and β , via the routine name.

2 Specification

```

real FUNCTION G01EFF(TAIL, G, A, B, IFAIL)
INTEGER                IFAIL
real                 G, A, B
CHARACTER*1           TAIL

```

3 Description

The lower tail probability for the gamma distribution with parameters α and β , $P(G \leq g)$, is defined by:

$$P(G \leq g; \alpha, \beta) = \frac{1}{\beta^\alpha \Gamma(\alpha)} \int_0^g G^{\alpha-1} e^{-G/\beta} dG, \quad \alpha > 0.0, \beta > 0.0$$

The mean of the distribution is $\alpha\beta$ and its variance is $\alpha\beta^2$. The transformation $Z = \frac{G}{\beta}$ is applied to yield the following incomplete gamma function in normalised form,

$$P(G \leq g; \alpha, \beta) = P(Z \leq g/\beta; \alpha, 1.0) = \frac{1}{\Gamma(\alpha)} \int_0^{g/\beta} Z^{\alpha-1} e^{-Z} dZ.$$

This is then evaluated using S14BAF.

4 References

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Parameters

- 1: TAIL – CHARACTER*1 *Input*
On entry: indicates whether an upper or lower tail probability is required.
 If TAIL = 'L', then the lower tail probability is returned, that is $P(G \leq g; \alpha, \beta)$.
 If TAIL = 'U', then the upper tail probability is returned, that is $P(G \geq g; \alpha, \beta)$.
Constraint: TAIL = 'L' or 'U'.
- 2: G – **real** *Input*
On entry: the value of the gamma variate, g .
Constraint: $G \geq 0.0$.
- 3: A – **real** *Input*
On entry: the parameter α of the gamma distribution.
Constraint: $A > 0.0$.

- 4: B – *real* *Input*
On entry: the parameter β of the gamma distribution.
Constraint: $B > 0.0$.
- 5: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1 . Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If IFAIL = 1, 2, 3 or 4 on exit, then G01EFF returns 0.0.

IFAIL = 1

On entry, TAIL \neq 'L' or 'U'.

IFAIL = 2

On entry, $G < 0.0$.

IFAIL = 3

On entry, $A \leq 0.0$,
 or $B \leq 0.0$.

IFAIL = 4

The solution did not converge in 600 iterations. See S14BAF. The probability returned should be a reasonable approximation to the solution.

7 Accuracy

The result should have a relative accuracy of *machine precision*. There are rare occasions when the relative accuracy attained is somewhat less than *machine precision* but the error should not exceed more than 1 or 2 decimal places. Note also that there is a limit of 18 decimal places on the achievable accuracy, because constants in S14BAF are given to this precision.

8 Further Comments

The time taken by the routine varies slightly with the input parameters g , α and β .

9 Example

The example program reads in values for several gamma distributions, computes and prints the lower probabilities for each case, until the end of data is reached.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G01EFF Example Program Text
*      Mark 14 Release.  NAG Copyright 1989.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
real           A, B, G, P
INTEGER         IFAIL
*      .. External Functions ..
real           G01EFF
EXTERNAL        G01EFF
*      .. Executable Statements ..
WRITE (NOUT,*) 'G01EFF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
WRITE (NOUT,*)
WRITE (NOUT,*)
+ 'Gamma deviate      Alpha      Beta      Lower tail prob.'
WRITE (NOUT,*)
20 READ (NIN,*,END=40) G, A, B
   IFAIL = 0
*
   P = G01EFF('Lower',G,A,B,IFAIL)
*
   WRITE (NOUT,99999) G, A, B, P
   GO TO 20
40 STOP
*
99999 FORMAT (1X,F9.2,F13.2,F9.2,F14.4)
END
```

9.2 Program Data

```
G01EFF Example Program Data
15.5  4.0  2.0      :G A B
 0.5  4.0  1.0      :G A B
10.0  1.0  2.0      :G A B
 5.0  2.0  2.0      :G A B
```

9.3 Program Results

```
G01EFF Example Program Results

Gamma deviate      Alpha      Beta      Lower tail prob.

15.50              4.00      2.00      0.9499
 0.50              4.00      1.00      0.0018
10.00              1.00      2.00      0.9933
 5.00              2.00      2.00      0.7127
```
